#被动技能【攻略组】10
item replace entity @a[tag=qishi,scores={skPlayer=10..19}] armor.head with minecraft:white_stained_glass{AttributeModifiers: [{Operation: 1, Amount: 0.033d, UUID: [I; 642185238, -1493802177, -417730070, -1081945800], Slot: head, AttributeName: "generic.movement_speed", Name: "generic.movementSpeed"}, {Operation: 0, Amount: +2, UUID: [I; 611081318, -1079049893, -363144022, -1343007698], Slot: head, AttributeName: "generic.armor_toughness", Name: "generic.armorToughness"}], display: {Name: '[{"text":"血盟10+","color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]'}} 1
#被动技能【攻略组】20
item replace entity @a[tag=qishi,scores={skPlayer=20..29}] armor.head with minecraft:light_blue_stained_glass{AttributeModifiers: [{Operation: 0, Amount: +4, UUID: [I; 611081318, -1079049893, -363144022, -1343007698], Slot: head, AttributeName: "generic.armor_toughness", Name: "generic.armorToughness"}, {Operation: 0, Amount: 2, UUID: [I; 569420852, 1915381635, -261401831, 725365584], Slot: head, AttributeName: "generic.armor", Name: "generic.armor"}, {Operation: 0, Amount: +0.4d, UUID: [I; 572104435, -1283295932, -311322200, 718778631], Slot: head, AttributeName: "generic.attack_speed", Name: "generic.attackSpeed"}], display: {Name: '[{"text":"血盟20+","color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]'}} 1
#被动技能【攻略组】30
effect give @a[tag=qishi,scores={skPlayer=30}] resistance 1 0 true
item replace entity @a[tag=qishi,scores={skPlayer=30}] armor.head with minecraft:blue_stained_glass{AttributeModifiers: [{Operation: 0, Amount: +1, UUID: [I; 621618659, -691985361, -338009859, 1357902007], Slot: head, AttributeName: "generic.attack_damage", Name: "generic.attackDamage"}, {Operation: 0, Amount: +0.6d, UUID: [I; 517498544, -1239604567, -452684790, -1739925444], Slot: head, AttributeName: "generic.attack_speed", Name: "generic.attackSpeed"}, {Operation: 1, Amount: 0.1d, UUID: [I; 585595511, -1406409195, -284670228, 516565569], Slot: head, AttributeName: "generic.movement_speed", Name: "generic.movementSpeed"}], display: {Name: '[{"text":"血盟30","color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]'}} 1
clear @a[tag=qishi,scores={skPlayer=..9}] white_stained_glass
clear @a[tag=qishi,scores={skPlayer=..19}] light_blue_stained_glass
clear @a[tag=qishi,scores={skPlayer=..29}] blue_stained_glass

#被动技能【攻略组】造成伤害或受到伤害+1
scoreboard players add @a[tag=qishi,scores={skShangHai=1..,skPlayer=0..29}] skPlayer 1
scoreboard players set @a[tag=qishi,scores={skShangHai=1..}] skShangHai 0
effect clear @a[tag=qishi,scores={skShouShang1=1..}] regeneration
scoreboard players add @a[tag=qishi,scores={skShouShang1=1..,skPlayer=0..29}] skPlayer 1
scoreboard players set @a[tag=qishi,scores={skShouShang1=1..}] skShouShang1 0
#被动技能【攻略组】战斗状态每2秒+1
scoreboard players add @a[tag=qishi,scores={skZhanDouState=1..,skCDPlayer1=..0,skPlayer=0..29}] skPlayer 1
scoreboard players set @a[tag=qishi,scores={skZhanDouState=1..,skCDPlayer1=..0,skPlayer=0..29}] skCDPlayer1 2
scoreboard players set @a[tag=qishi,scores={skZhanDouState=0}] skCDPlayer1 2
#被动技能【攻略组】非战斗状态每秒-1
scoreboard players remove @a[tag=qishi,scores={skZhanDouState=0,skCDPlayer=1..,skPlayer=1..}] skPlayer 1
scoreboard players reset @a[tag=qishi,scores={skZhanDouState=0,skCDPlayer=1..}] skCDPlayer

#被动技能【神圣盾】
scoreboard players set @a[tag=qishi,scores={skShield0=1..}] skCDEntity1 1
scoreboard players set @a[tag=qishi,scores={skShield0=1..}] skShield0 0
#被动技能【神圣盾】盾击
execute as @a[tag=qishi,scores={skCDEntity1=1..,skIron_sword0=1..}] at @s rotated ~ 0 positioned ^ ^ ^3.1 run effect give @p[gamemode=!spectator,distance=0..3] blindness 2 0 true
execute as @a[tag=qishi,scores={skCDEntity1=1..,skIron_sword0=1..}] at @s rotated ~ 0 positioned ^ ^ ^3.1 run effect give @p[gamemode=!spectator,distance=0..3] slowness 1 4 true
execute at @a[tag=qishi,scores={skCDEntity1=1..,skIron_sword0=1..}] rotated ~ 0 positioned ^ ^ ^3.1 as @p[gamemode=!spectator,distance=0..3] at @s rotated ~ 75 run tp @s ~ ~ ~ ~ ~

#武器技能【神圣剑】
execute as @a[tag=qishi,scores={skIron_sword0=1..}] at @s rotated ~ 0 positioned ^ ^ ^3.1 run effect clear @a[distance=0.1..3] regeneration
effect give @a[tag=qishi,scores={skIron_sword0=1..}] regeneration 4 1 true
scoreboard players set @a[tag=qishi,scores={skIron_sword0=1..}] skIron_sword0 0

#被动技能【GM权限】首次触发
execute as @a[tag=qishi,scores={qishi1=0,skHealth=..6}] run tellraw @a [{"selector":"@s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"：GM权限","color":"light_purple","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"已开启","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"—希兹克利夫","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
scoreboard players set @a[tag=qishi,scores={qishi1=0,skHealth=..6}] skCDEntity 0
effect give @a[tag=qishi,scores={qishi1=0,skHealth=..6}] minecraft:health_boost 1 19
execute as @a[tag=qishi,scores={qishi1=0,skHealth=..6}] run item replace entity @s armor.feet with minecraft:leather_boots{Tags: ["GM权限"], AttributeModifiers: [{Operation: 0, Amount: -10, UUID: [I; 686034547, -2070329382, -413911379, -773638537], Slot: feet, AttributeName: "generic.max_health", Name: "generic.maxHealth"}, {Operation: 0, Amount: +2, UUID: [I; 681364162, 985683493, -266029090, -1905452485], Slot: feet, AttributeName: "generic.armor", Name: "generic.armor"}], display: {Name: '[{"text":"GM权限修正中","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]', color: 6087911}, Unbreakable: 1, Enchantments: [{id: "minecraft:binding_curse", lvl: 1}]} 1
execute at @a[tag=qishi,scores={qishi1=0,skHealth=..6}] run particle heart ~ ~0.5 ~ 0.2 0.2 0.2 0.01 6 normal @a
effect give @a[tag=qishi,scores={qishi1=0,skHealth=..6}] instant_health 1 3
scoreboard players set @a[tag=qishi,scores={qishi1=0,skHealth=..6}] qishi1 1

execute as @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] run tellraw @a [{"selector":"@s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"：GM权限：","color":"blue","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"非系统性定格修正中","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"—希兹克利夫","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
execute at @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] run particle heart ~ ~0.5 ~ 0.2 0.2 0.2 0.01 6 normal @a
effect give @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] minecraft:health_boost 1 19
execute as @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] run item replace entity @s armor.feet with minecraft:leather_boots{Tags: ["GM权限"], AttributeModifiers: [{Operation: 0, Amount: -20, UUID: [I; 686034547, -2070329382, -413911379, -773638537], Slot: feet, AttributeName: "generic.max_health", Name: "generic.maxHealth"}, {Operation: 0, Amount: +2, UUID: [I; 681364162, 985683493, -266029090, -1905452485], Slot: feet, AttributeName: "generic.armor", Name: "generic.armor"}], display: {Name: '[{"text":"GM权限修正中","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]', color: 6087911}, Unbreakable: 1, Enchantments: [{id: "minecraft:binding_curse", lvl: 1}]} 1
effect give @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] instant_health 1 3
scoreboard players set @a[tag=qishi,scores={qishi1=1,skCDEntity=15..}] skCDEntity 0

clear @a[tag=qishi,nbt=!{ActiveEffects:[{Id:21b}]}] leather_boots
item replace entity @a[tag=qishi,nbt=!{ActiveEffects: [{Id: 21b}]}] armor.feet with iron_boots


#C09希兹克利夫（骑士）
execute as @a[tag=qishi,scores={qishi1=0}] run title @s actionbar [{"text":"血盟：","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skPlayer","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"/","color":"dark_red","bold":true,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"30","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
execute as @a[tag=qishi,scores={qishi1=1}] run title @s actionbar [{"text":"GM权限","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"即将到来：","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skCDEntity","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"s","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":" 血盟：","color":"dark_red","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"score":{"objective":"skPlayer","name":"@s"},"color":"aqua","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"/","color":"dark_red","bold":true,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false},{"text":"30","color":"gold","bold":false,"italic":false,"underlined":false,"strikethrough":false,"obfuscated":false}]
